

Computation versus Programming

Last time, we talked about **computation**

$(+ 1 (* 2 3)) \rightarrow (+ 1 6) \rightarrow 7$

Programming?

Make a wanted poster...



```
(define (maybe-wanted who wanted-who)
  (cond
    [(image=? who wanted-who)
     (above (text "WANTED" 32 "black") who)]
    [else
     who]))
```

We somehow wrote the function in one big, creative chunk.

Programming

Today: ***How to Design Programs***

- Programming always requires creativity
- But a design rules can guide and focus creativity
- We'll start with a simple recipe
- As the course progresses, we'll expand the recipe

Design Recipe I

Data

- Understand the input data: `num`, `bool`, `sym`, or `image`

Contract, Purpose, and Header

- Describe (but don't write) the function

Examples

- Show what will happen when the function is done

Body

- The most creative step: implement the function body

Test

- Run the examples

Data

Choose a representation suitable for the function input

- Fahrenheit degrees → `num`
- Grocery items → `string`
- Faces → `image`
- Wages → `num`
- ...

Handin artifact: **none** for now

Contract, Purpose, and Header

Contract

Describes input(s) and output data

- `f2c : num -> num`
- `is-milk? : string -> bool`
- `wearing-glasses? : image image image -> bool`
- `netpay : num -> num`

Handin artifact: a comment

```
; f2c : num -> num  
; is-milk? : string -> bool
```

Contract, Purpose, and Header

Purpose

Describes, in English, what the function will do

- Converts F-degrees **f** to C-degrees
- Checks whether **s** is a string for milk
- Checks whether **p2** is **p1** wearing glasses **g**
- Computes net pay (less taxes) for **n** hours worked

Handin artifact: a comment after the contract

```
; f2c : num -> num  
; Converts F-degrees f to C-degrees
```

Contract, Purpose, and Header

Header

Starts the function using variables that are mentioned in purpose

- `(define (f2c f))`
- `(define (is-milk? s))`
- `(define (wearing-glasses? p1 p2 g))`
- `(define (netpay n))`

Check: function name and variable count match contract

Handin artifact: as above, but absorbed into implementation

```
; f2c : num -> num
; Converts F-degrees f to C-degrees
(define (f2c f) ....)
```

Examples

Show example function calls and result

```
(check-expect (f2c 32) 0)
(check-expect (f2c 212) 100)

(check-expect (is-milk? "milk") true)
(check-expect (is-milk? "apple") false)
```

Check: function name, argument count and types match contract

Handin artifact: as above, after header/body

```
; f2c : num -> num
; Converts F-degrees f to C-degrees
(define (f2c f) ...)
(check-expect (f2c 32) 0)
(check-expect (f2c 212) 100)
```


Body

Fill in the body under the header

```
(define (f2c f)
  (* (- f 32) 5/9))


(define (is-milk? s)
  (string=? s "milk"))
```

Handin artifact: complete at this point

```
; f2c : num -> num
; Converts F-degrees f to C-degrees
(define (f2c f)
  (* (- f 32) 5/9))
(check-expect (f2c 32) 0)
(check-expect (f2c 212) 100)
```

Test

Click **Execute** — examples serve as tests



The screenshot shows the DrRacket IDE window titled "f2c.rkt - DrRacket". The menu bar includes "f2c.rkt", "(define ...)", "Step", "Check Syntax", "Run", and "Stop". The code editor contains the following Racket code:

```
; f2c : num -> num
; Converts F-degrees f to C-degrees
(define (f2c f)
  (* (- f 32) 5/9))
(check-expect (f2c 32) 0)
(check-expect (f2c 212) 100)
```

The output area displays the following text:

```
Welcome to DrRacket, version 5.0.1.3--2010-08-25(f13dcc2/g) [3m].
Language: Beginning Student; memory limit: 256 MB.
Both tests passed!
>
```

The status bar at the bottom shows "Beginning Stud...", "1:0", and "126.67 MB".

Design Recipe - Each Step Has a Purpose

Data

- Shape of input data will drive the implementation

Contract, Purpose, and Header

- Provides a first-level understanding of the function

Examples

- Gives a deeper understanding and exposes specification issues

Body

- The implementation is the whole point

Test

- Evidence that it works

Design Recipe FAQ

- Do I have to use the recipe when the function seems obvious?
 - **Yes.**
- Will my grade suffer if I don't hand in recipe artifacts?
 - **Yes,** except for HW 0
- Isn't the recipe just a lot of obnoxious busy work?
 - **No.** It's a training exercise.

As programs become more complex in the next few weeks, the design recipe will prove more helpful.

If you don't learn to use the recipe now, you'll be stuck having to learn both the recipe and other concepts later on.